

“Basics of microcontroller”

Lecture 5.

*Interrupt, serial input and output,
AD0 - AD7 port features*

Lecturer: Onon Otgonbaatar, MD.

ГАР БА ДЭЛГЭЦТЭЙ АЖИЛЛАХ ҮЙЛДЛҮҮД

Ихэнх програмууд гар, диск, модем зэргээс оролт шаарддаг ба дэлгэц, принтер, диск үрүү мэдээлэл гаргадаг. Дэлгэц болон гар дээрх өгөгдлүүд нь ASCII форматаар явагдана.

INT (interrupt) команд нь оролт ба гаралтад чухал үүрэг гүйцэтгэнэ.

BIOS уруу удирдлага дамжуулдаг INT 10H тасалдлыг ашиглан гар ба дэлгэцний бүх шаардлагатай үйлдлүүдийг гүйцэтгэж болно. Илүү төвөгтэй үйлдлүүдэд илүү өндөр түвшний тасалдал болох DOS-д удирдлага дамжуулдаг INT 21H-ийг ашиглана. Жишээ нь: Гарнаас оруулах өгөгдөл нь тэмдэгтүүдийн тоо, хамгийн ихдээ оруулах тооны хязгаарыг шалгах, return тэмдэгтийг шалгах зэргийг шаарддаг. DOS INT 21H функц нь эдгээр нэмэлт үйлдлүүдийг агуулах ба BIOS уруу автоматаар дамжуулна.

Тасалдал (INTERRUPT)

Тасалдал гэдэг нь үйлчилгээ хүссэнийн улмаас явагдаж буй үйл ажиллагааг зогсоон, урд өмнө нь бэлтгэгдсэн процедурыг явуулахыг хэлнэ. Уг процедур дуусахад таслагдсан ажиллагааг үргэлжлүүлнэ.

Микропроцессорт 255 (FF) ширхэг тасалдал байх ба санах ойн эхний 1024 байт нь (00000 – 003FF хаягтай) тасалдлын вектор хаягуудад нөөцлөгдөнө. Энэ вектор хаяг гэдэг нь 1 тасалдлын хувьд 4 байт ба эхний 2 байт нь тасалдлийн үйлчилгээний шилжлэг хаягийг, дараагийн 2 байт нь сегмент хаягийг агуулна.

Микропроцессорын тасалдлуудыг 2 ангилж болно.

- системийн тасалдал: hardware (гар, диск г.м) , software (BIOS 10-1F тасалдлын дугаартай, DOS 20-3F тасалдлын дугаартай)
- хэрэглэгчийн тасалдал: програмистууд өөрснөө зохионо. (60-6F тасалдлын дугаар дээр)

Системийн hardware тасалдлуудыг үзье.

	Тасалдал	Сис/дугаар	Хаяг	
0	Timer (8253)		8	20-23 өндөр урьтамжтай
1	Keyboard		9	24-27
2	Нөөц		A	28-2B
3	Communication		B	2C-2F
4	Communication		C	30-33
5	Hard disk driver		D	34-37
6	Floppy disk driver		E	38-3B
7	Printer		F	3C-3F бага урьтамжтай

Системийн программчилагддаг тоолуур элемент нь тодорхой хугацаа бүрт 8 дахь тасалдалыг үүсгэн, BIOS-уруу вектор хаягаараа үсрэн тасалдлын үйлчилгээг үзүүлж, цаг хугацааг зохицуулдаг.

Гарнаас товч дарагдах бүрт 9 дэх тасалдал үүсдэг. Мөн BIOS тасалдлан үйлчилгээг үзүүлж, портоор гарнаас кодыг авч, ASCII кодоод хөрвүүлэн, буферт хадгалдаг.

Тасалдын команд INT

INT команд нь программыг таслан, DOS буюу BIOS-д удирдлагыг дамжуулна. Голдуу оруулах, гаргах үйлдлүүдийг гүйцэтгэнэ. Бүх тасалдлууд программаас гарахдаа программдаа эргэж ороход шаардлагатай зүйлсийг хийнэ. INT команд нь доорхийг хийнэ.

- Стекийн заагчийн утгыг 2-оор хорогдуулах ба флаг регистрийн утгыг стекд (push) хийнэ.
- TF ба IF флаг битүүдийг цэвэрлэнэ.
- Стекийн заагчийн утгыг 2-оор хорогдуулах ба CS регистрийн утгыг стекд (push) хийнэ.
- Стекийн заагчийн утгыг 2-оор хорогдуулах ба командын заагч (IP) регистрийн утгыг стекд (push) хийнэ.
- Гүйцэтгэх ёстой үйлдлээ хийнэ.
- Регистрүүдээ стекээс (pop) авах ба INT командын дараагийн команд уруу ирж программаа үргэлжлүүлнэ.

Эдгээр үйлдлүүд нь бүхэлдээ автоматаар хийгдэх бөгөөд программист нь стек сегментээ л мэдээлэл хадгалахад хангалттай хэмжээтэйгээр зарлах хэрэгтэй юм.

BIOS INT 10H: гар ба дэлгэцийн үйл ажиллагаанд хэрэглэнэ.

DOS INT 21H: гаралтыг дэлгэц дээр харуулах, оролтыг хүлээж авахад хэрэглэгдэнэ.

ДЭЛГЭЦ

Курсорыг тавих

Дэлгэц гэдэг нь курсорыг дурын байрандаа тавиж болох хаяглах боломжтой цэгүүд юм. Энгийн видео монитор нь 25 мөр (0-ээс 24 хүртэл хаяглагддаг), 80 багана (0-ээс 79 хүртэл хаяглагддаг)-тай байна. Курсорын зарим байрлалуудыг дурдвал:

Байрлал	Мөр (decimal)	Багана (decimal)	Мөр (hex)	Багана (hex)
Зүүн дээд өнцөг	00	00	00	00
Баруун дээд өнцөг	00	79	00	4F
Дэлгэцийн төв	12	39 / 40	0C	27 / 28
Зүүн доод өнцөг	24	00	18	00
Баруун доод өнцөг	24	79	18	4F

Доорх программ нь курсорыг 5-р мөр, 12-р багана дээр байрлуулна.

```
MOV    AH, 02      ; курсор тавих хүсэлтийг 10h тасалдалд тавина.
MOV    BH, 00      ; дэлгэцний дугаар 0
MOV    DH, 05      ; 5-р мөр
MOV    DL, 12      ; 12-р багана
INT    10H         ; тасалдал – BIOS уруу гарна.
```

DX регистрт мөр, баганын дугаар байх учир доорх байдлаар команд өгч болно.
MOVDX, 050CH – мөр 5, багана 12

Дэлгэцийг цэвэрлэх

Дэлгэц дээр бичигдсэн зүйлс нь дахин юм бичих буюу гүйлгэх хүртэл хэвээрээ л хадгалагддаг. Шинээр программ бичихдээ программист дэлгэцээ цэвэрлэхийн тулд DX регистрт эхлэх мөр, баганын дугаарыг өгч, BH регистрт 07 гэсэн утга өгөн, AX регистрт 0600h утга олгоно.

MOV	AX, 0600H	; AH 06 – хоосолно, AL 00- бүхэл дэлгэц
MOV	BH, 07	; энгийн атрибут (хар ба цагаан)
MOV	CX, 0000	; хамгийн дээд мөр, зүүн багана
MOV	DX, 184FH	; хамгийн доод мөр, баруун багана
INT	10H	; тасалдал – BIOS уруу гарна.

Энэ программ нь хар цагаан дэлгэцэнд аюулгүй боловч харин зарим өнгөт дэлгэцэнд осолтой байж болно.

Дэлгэцэнд мэдээлэл гаргах

Өгөгдлийн талбарт дэлгэцэнд харуулах мэдээллээ хадгална. Мэдээллийн төгсгөлийг \$ тэмдэгтээр танина. Жишээ нь:

NAMPRMP	DB	'Customer name?', '\$'
...		
MOV	AH, 09	; дэлгэцийг хүсэх тасалдал
LEA	DX, NAMPRMP	; мэдээллийн хаягийг авна.
INT	21H	; DOS-ын тасалдал

NAMPRMP	DB	'Customer name?\$' гэж бичсэн ч болно. Эсвэл
NAMPRMP	DB	'Customer name?'
DB	'\$'	гэж 2 мөрөөр бичсэн ч болно.

\$-ийн тэмдэгтийг дэлгэц дээр гаргахдаа энэ командыг хэрэглэж болохгүй. Мөн \$-ийн тэмдгийг бичилгүй орхивол тэмдэгтүүдийн төгсгөлийг мэдэхгүй болж, \$-ийн тэмдэгт тааралдтал тэмдэгтүүдийг дэлгэцэнд хэвлэнэ.

LEA DX, NAMPRMP команд нь DX-д бичих тэмдэгтүүдийн эхлэлийн offset (шилжлэг) хаягийг авна. DOS нь DS регистрийн утгыг аваад DS:DX гэсэн хос регистрээр хаяглан өгөгдлөө олж авна.

DOS-ийн 21h тасалдалын өргөтгөсөн хувилбар

- AH регистрт дуудах функцийнхээ дугаарыг өгнө.
- 3FH нь оролт, 40H нь гаралтад хэрэглэгдэнэ.
- CX регистрт унших буюу дэлгэцэнд харуулах байтынхаа тоог тавина.
- DX регистрт оролт буюу гаралтын талбарын хаягийг ачаална.

Амжилттай үйлдлийн дараа Carry флагийг цэвэрлэх (0) ба AX регистрт оруулсан буюу дэлгэцэнд хэвлэсэн тэмдэгтүүдийн тоог буцаана.

Амжилтгүй үйлдлийн дараа Carry флагийг тавих (1) ба AX регистрт алдааны код 6 гэсэн тоог буцаана.

Зарим стандарт төхөөрөмжүүдэд хандахдаа тэднийг нээхгүйгээр шууд хандаж болно. Өргөтгөсөн DOS-ын орчинд оролт гаралтын үйлдэл хийхдээ BX регистрт доорх файлын мэдээг тавьж өгнө.

0 – оролт, голдуу гарнаас (CON)

1 – гаралт, голдуу дэлгэц (CON)

2 – алдааны гаралт, голдуу дэлгэц (CON)

3 – туслах төхөөрөмж (AUX)

4 – принтер (LPT1 ба PRN)

Гарнаас өгөгдлийг хүлээж авах доорх арга нь DOS-ын өргөжүүлсэн хувилбар юм.

```
INAREA DB 20 DUP(' '); өгөгдөл оруулах талбар
```

```
....
```

```
MOV AH, 3FH ; оролт хүсэх
```

```
MOV BX, 00 ; гарыг эзэмшинэ.
```

```
MOV CX, 20 ; хамгийн ихдээ 20 тэмдэгт
```

```
LEA DX, INAREA ; оролтын талбарын хаяг
```

```
INT 21H ; DOS –ыг дуудна.
```

Хэрэв *PC_My_Computer* гэсэн тэмдэгтүүдийг гарнаас оруулбал INAREA талбарт

PC_My_Computer, 0DH, 0AH

гэсэн мэдээлэл байна. 0DH нь гарнаас оруулсан return тэмдэгт, 0AH нь оруулаагүй line feed тэмдэгт байна.

Дэлгэцэнд өгөгдлийг хэвлэх доорх арга нь DOS-ын өргөжүүлсэн хувилбар юм.

```
DISAREA DB 'Computer science'
```

```
....
```

```
MOV AH, 40H ; дэлгэцэнд хэвлэхийг хүсэх
```

```
MOV BX, 01 ; гаралтыг эзэмшинэ.
```

```
MOV CX, 20 ; хамгийн ихдээ 20 тэмдэгт
```

LEA DX, DISAREA ; дэлгэцийн талбарын хаяг
 INT 21H ; DOS –ыг дуудна.

DX регистрт дэлгэцэнд харуулах *Computer science* гэсэн үгний байрлаж буй DISAREA талбарын шилжлэг хаяг байрлана.

Return тэмдэгтийг арилгах

Оролтын тэмдэгтүүдийг притерээр хэвлэх, дискэнд бичих зэрэг зорилгоор заримдаа return тэмдэгт (0Dh)-ийг хоосон зай (20h)-гаар солих хэрэг гардаг. Орсон тэмдэгтүүдийн тоо NAMELEN 5 байгаа үед return тэмдэгтийн байрлал NAMEFLD+5 гэдэг нь тодорхой болно.

MOV BH, 0
 MOV BL, NAMELEN ; BX ← 0005
 MOV NAMEFLD[BX], 20H ; return тэмдэгтийг хоослоно.

Оролтын талбарыг цэвэрлэх

Шинээр оруулсан өгөгдлүүд нь өмнөх утгаа даран байрлах ба өөр утга оруултал тэндээ байсаар байна. Жишээ нь:

BROWN	14	05	42	52	4F	57	4E	0D	20	20	20	20
HAMILTON	14	08	48	41	4D	49	4C	54	4F	4E	0D	20
ADAMS	14	05	41	44	41	4D	53	0D	4F	4E	0D	20

Эндээс харахад HAMILTON гэсэн нэрний сүүлийн O, N үсгүүд нь хадгалагдан үлдэж байна. Иймд шинэ нэр оруулахын өмнө NAMEFLD-ийг доорх байдлаар цэвэрлэж болно.

MOV CX, 20 ; 20 давталтын тоолуурын утгыг олгоно.
 MOV SI, 0000 ; нэрний талбарын эхлэлийн байрлал 0000
 B30:
 MOV NAMEFLD[SI], 20H ; 1 хоосон тэмдэгт оруулах
 INC SI ; дараагийн тэмдэгтийг заана.
 LOOP B30 ; 20 удаа давтана.

SI регистрийн оронд DI эсвэл BX регистрийг ашиглаж болно. Цөөн давталт хийхийн тулд NAMEFLD-ийг доорх байдлаар word утгаар цэвэрлэж болно.

;-----WORD
 MOV CX, 10 ; 20 давталтын тоолуурын утгыг олгоно.
 LEA SI, NAMEFLD ; нэрний талбарын эхлэх хаягийг авна.

B30:

```
MOV    WORD PTR[SI], 2020H    ; 2 хоосон тэмдэгт олгоно.
INC    SI                      ; 2 байгаар
INC    SI                      ; нэмэгдүүлнэ.
LOOP   B30                     ; 10 удаа давтана.
```

SI регистрийг ашиглалгүй MOV WORD PTR[NAMEFLD], 2020H гэж бичвэл алдаа болно.

Жишээ программ

Гарнаас оруулсан нэрийг дэлгэцэнд харуулах программ

```
TITLE    CTRNAME Accept name and center on screen
.model   small
.stack   100h
.data
    NAMEPAR    LABEL    BYTE    ; параметр лист
    MAXLEN     DB        20      ; нэрийн максимум урт
    NAMELEN    DB        ?       ; оруулсан тэмдэгтийн тоо
    NAMEFLD    DB        20 DUP(' '), '$' ; оруулсан нэрийг байрлуулах талбар
    PROMPT     DB        'NAME ?', '$'

.code
BEGIN    PROC    FAR
    MOV     AX, @DATA
    MOV     DS, AX
    MOV     ES, AX
    CALL    DISCLEAR    ; Дэлгэц цэвэрлэнэ.
ALOOP:
    MOV     DX, 0000    ; Курсорыг 00.00
    CALL    SETCUR     ; байрлалд тавина.
    CALL    DISPROMPT  ; Prompt-ийг дэлгэцэнд хэвлэнэ.
    CALL    INPT       ; Нэрийг оруулна.
    CALL    DISCLEAR   ; Дэлгэцийг цэвэрлэнэ.
    CMP     NAMELEN, 00    ; Нэр орсон уу?
    JE     ENDING        ; Үгүй бол гарна
    CALL    SETBELL     ; Хонх тавина.
    CALL    DISNAME     ; Нэрийг дэлгэцийн голд хэвлэнэ.
    JMP     ALOOP
ENDING:
    RET                ; DOS уруу буцна.
BEGIN    ENDP
```

;---PROMPT-ийг дэлгэцэнд хэвлэх

DISPROMPT PROC NEAR

```
MOV    AH, 09           ; Дэлгэцэнд хэвлэх хүсэлт
LEA    DX, PROMPT
INT    21H
RET
```

DISPROMPT ENDP

;---Оруулсан нэрийг хүлээж авах

INPT PROC NEAR

```
MOV    AH, 0AH         ; Оролт хүсэх
LEA    DX, NAMEPAR
INT    21H
RET
```

INPT ENDP

;---Хонх тавих

SETBELL PROC NEAR

```
MOV    BH, 00           ; Return (0D) тэмдгийг хонхоор
MOV    BL, NAMELEN      ;           (07) солино.
MOV    NAMEFLD[BX], 07
MOV    NAMEFLD[BX+1], '$' ; Дэлгэцийн '$'-ийг хэвлэнэ.
RET
```

SETBELL ENDP

;---Дэлгэцийн төвд нэрийг хэвлэх

DISNAME PROC NEAR

```
MOVDL, NAMELEN ;
SHR    DL, 1           ; нэрийн уртыг 2-т хуваана.
NEG    DL              ; тэмдгийг урвуулна.
ADD    DL, 40          ; 40-ийг нэмнэ.
MOV    DH, 12         ; Голын мөр
CALL   SETCUR         ; Курсорыг байрлуулна.
MOV    AH, 09
LEA    DX, NAMEFLD    ; Нэрийг хэвлэнэ.
INT    21H
RET
```

DISNAME ENDP

;---Дэлгэц цэвэрлэх

DISCLEAR PROC NEAR

```
MOV    AX, 0600H       ; Дэлгэц цэвэрлэх хүсэлт
MOV    BH, 30          ; Өнгө тавина. (хар цагаан бол 07)
MOV    CX, 0000H ;     00,00 байрлалаас
MOV    DX, 184FH ;     24,79 хүртэл
INT    10H            ; BIOS-ийг дуудна.
```

```

RET
DISCLEAR ENDP
;----Курсорыг байрлуулах
SETCUR PROC NEAR          ; DX-д оролтоо авна.
    MOV     AH, 02         ; Курсор тавих хүсэлт
    MOV     BH, 00         ; 0-р хуудас
    INT     10H           ; BIOS-ийг дуудна.
    RET
SETCUR ENDP
END     BEGIN

```

Уг программ нь хэрэглэгчийг гарнаас нэр оруулахыг хүснэ. Дараа нь нэрийг дэлгэцийн голд хэвлэх ба дуу гаргана. Энэ программ хэрэглэгчийг Return товчлуурыг дартал үргэлжилнэ. Жишээ нь хэрэглэгч BOLD AMAR гэж оруулбал:

- Нэрийн урт 09-ийг 2-т хуваахад 4 болно.
- Энэ утгыг 40-өөс хасахад 36 болно.

DISNAME процедур дахь SHR DL, 1 команд нь нэрийн уртыг 2-т хуваана. 0000.1001 маань 0000.0100 болно. NEG DL команд нь тэмдгийг урвуулан +4 гэсэн тоог -4 болгоно. ADD DL, 40 команд нь эхлэх баганын дугаарыг зааж 40-4=36-г тодорхойлно. Ингээд 12-р мөр, 36-р өагана дээрээс эхлэн нэр хэвлэгдэнэ.

SETBELL процедур дахь нэрний дараа хонх (07)-ыг

```
MOV BL, NAMELEN
```

```
MOV NAMEFLD[BX], 07
```

командуудаар оруулж өгнө. BX регистрт нэрийн урт 09-ийг хийхэд, NAMEFLD+9 хаяг дахь Return тэмдэгтийн оронд 07-г хийж өгнө. Сүүлд нь '\$' -ын тэмдэг буюу төгсгөлийг оруулснаар нэрийг хэвлэхийн зэрэгцээ мөн speaker-ээр дуу гаргана.

Хэрэв оруулах тэмдэгтийн хамгийн их тооноос хэтэрвэл speaker нь дуу гаргах ба зөвхөн Return тэмдэгтийг хүлээж авна. Мөн зөвхөн Return тэмдэгтийг оруулах юм бол зөвхөн энэ тэмдэгтийг хүлээж аваад, параметр листэд нэрийн уртыг 0 гэж доорх байдлаар авна.

Parameter list (hex): 14 | 00 | 0D | ...

Return тэмдэгтээр программ оролт дууссанг мэднэ.

Оруулсан утгыг өөр бичлэгт нэмж оруулах, хүснэгтэд хадгалах, дискэнд бичих гэх мэт шаардлагууд гарч болно. Ийм үед Return (0D) тэмдэгтийг хоосон (20h) тэмдэгтээр солих хэрэгтэй болно. Үүнийг хийхдээ доорх байдлаар хийж болно.

```
MOV     BH, 00
```

```
MOV     BL, NAMELEN          ; Нэрийн төгсгөлийн хаягийг авна.
```

```
MOV     NAMEFLD[BX], 20H     ; хоосон тэмдэгтээр солино.
```

Textbook

1. Peter Abel, "IBM PC Assembler language and programming", USA, 1987
2. Jim Mischel, "Macro magic with Turbo Assembler", USA, 1993
3. William C. Runnion, "Structured programming in Assembly Language for the IBM PC", Boston, 1988
4. Thomas A. Wadlow, "Memory resident programming on the IBM PC", USA, 1987
5. Robert S. Lai, "Writing MS-DOS device drivers", USA, 1987
6. E. Majisuren, "IBM assembly language", MGL, 2003
7. Muhammad Ali Mazidi, Janice Gillispie Mazidi, Rolin D. McKinlay, "The 8051 microcontroller and Embedded Systems Using Assembly and C", USA, 2007